```
> with(GraphTheory);
```

[*AcyclicPolynomial*, *AddArc*, *AddEdge*, *AddVertex*, *AdjacencyMatrix*, *AllPairsDistance*, *Arrivals*, *ArticulationPoints*, *BellmanFordAlgorithm*, *BiconnectedComponents*, *BipartiteMatching*, *Blocks*, *CartesianProduct*, *CharacteristicPolynomial*, *ChromaticIndex*, *ChromaticNumber*, *ChromaticPolynomial*, *CircularChromaticIndex*, *CircularChromaticNumber*, *CircularEdgeChromaticNumber*, *CliqueNumber*, *CompleteGraph*, *ConnectedComponents*, *Contract*, *ConvertGraph*, *CopyGraph*, *CycleBasis*, *CycleGraph*, *Degree*, *DegreeSequence*, *DelaunayTriangulation*, *DeleteArc*, *DeleteEdge*, *DeleteVertex*, *Departures*, *Diameter*, *Digraph*, *DijkstrasAlgorithm*, *DiscardEdgeAttribute*, *DiscardGraphAttribute*, *DiscardVertexAttribute*, *DisjointUnion*, *Distance*, *DrawGraph*, *DrawNetwork*, *DrawPlanar*, *EdgeChromaticNumber*, *EdgeConnectivity*, *Edges*, *ExportGraph*, *FlowPolynomial*, *FundamentalCycle*, *GetEdgeAttribute*, *GetEdgeWeight*, *GetGraphAttribute*, *GetVertexAttribute*, *GetVertexPositions*, *Girth*, *Graph*, *GraphComplement*, *GraphEqual*, *GraphJoin*, *GraphNormal*, *GraphPolynomial*, *GraphPower*, *GraphRank*, *GraphSpectrum*, *GraphUnion*, *GreedyColor*, *HasArc*, *HasEdge*, *HighlightEdges*, *HighlightSubgraph*, *HighlightTrail*, *HighlightVertex*, *HighlightedEdges*, *HighlightedVertices*, *ImportGraph*, *InDegree*, *IncidenceMatrix*, *IncidentEdges*, *IndependenceNumber*, *InducedSubgraph*, *IsAcyclic*, *IsBiconnected*, *IsBipartite*, *IsClique*, *IsConnected*, *IsCutSet*, *IsDirected*, *IsEdgeColorable*, *IsEulerian*, *IsForest*, *IsGraphicSequence*, *IsHamiltonian*, *IsIntegerGraph*, *IsIsomorphic*, *IsNetwork*, *IsPlanar*, *IsRegular*, *IsStronglyConnected*, *IsTournament*, *IsTree*, *IsTwoEdgeConnected*, *IsVertexColorable*, *IsWeighted*, *IsomorphicCopy*, *KruskalsAlgorithm*, *LineGraph*, *ListEdgeAttributes*, *ListGraphAttributes*, *ListVertexAttributes*, *MakeDirected*, *MakeWeighted*, *MaxFlow*, *MaximumClique*, *MaximumDegree*, *MaximumIndependentSet*, *MinimalSpanningTree*, *MinimumDegree*, *Mycielski*, *Neighborhood*, *Neighbors*, *NonIsomorphicGraphs*, *NumberOfEdges*, *NumberOfSpanningTrees*, *NumberOfVertices*, *OddGirth*, *OutDegree*, *PathGraph*, *PermuteVertices*, *PlaneDual*, *PrimsAlgorithm*, *RandomGraphs*, *RankPolynomial*, *RelabelVertices*, *SeidelSpectrum*, *SeidelSwitch*, **(1)** *SequenceGraph, SetEdgeAttribute, SetEdgeWeight, SetGraphAttribute, SetVertexAttribute, SetVertexPositions, ShortestPath, SpanningPolynomial, SpanningTree, SpecialGraphs, StronglyConnectedComponents, Subdivide, Subgraph, TensorProduct, TopologicSort, TravelingSalesman, TreeHeight, TuttePolynomial, TwoEdgeConnectedComponents, UnderlyingGraph, VertexConnectivity, Vertices, WeightMatrix*]

```
> with(SpecialGraphs);
```

[*AntiPrismGraph, CageGraph, ClebschGraph, CompleteBinaryTree, CompleteKaryTree, CoxeterGraph, DesarguesGraph, DodecahedronGraph, DoubleStarSnark, DyckGraph, FlowerSnark, FosterGraph, GeneralizedBlanusaSnark, GeneralizedHexagonGraph, GeneralizedPetersenGraph, GoldbergSnark, GridGraph, GrinbergGraph, GrotzschGraph, HeawoodGraph, HerschelGraph, HoffmanSingletonGraph, HypercubeGraph, IcosahedronGraph, KneserGraph, LCFGraph, LeviGraph, McGeeGraph, MobiusKantorGraph, OctahedronGraph, OddGraph, PappusGraph, PayleyGraph, PetersenGraph, PrismGraph, RobertsonGraph, ShrikhandeGraph, SoccerBallGraph, StarGraph, SzekeresSnark, TetrahedronGraph, ThetaGraph, TorusGridGraph, Tutte8CageGraph, WebGraph, WheelGraph*] **(2)**

```
> W5 := WheelGraph(5);
```

W5 := *Graph 1: an undirected unweighted graph with 6 vertices and 10 edge(s)* **(3)**

```
> DrawGraph(W5);
```




```
> IsVertexColorable(W5,3);
```

false **(4)**

```
> P := PetersenGraph();
```

P:= Graph 2: an undirected unweighted graph with 10 vertices and 15 edge(s) **(5)**

```
> DrawGraph(P);
```



The Petersen graph is 3 colorable.

```
> IsVertexColorable(P,3,'C');
```

true **(6)**

```
> C;
```

[[1, 3, 8, 10], [2, 4, 6], [5, 7, 9]] **(7)**

```
> HighlightVertex(P,C[1],red);
```

```
> HighlightVertex(P,C[2],green);
> DrawGraph(P);
```




```
> Vertices(W5);
```

$$[0, 1, 2, 3, 4, 5] \quad \textbf{(8)}$$

```
> seq( x[u]^k-1, u=Vertices(W5) );
```

$$x_0^k - 1,\ x_1^k - 1,\ x_2^k - 1,\ x_3^k - 1,\ x_4^k - 1,\ x_5^k - 1 \quad \textbf{(9)}$$

```
> Edges(W5);
```

$$\{\{0, 1\}, \{0, 2\}, \{0, 3\}, \{0, 4\}, \{0, 5\}, \{1, 2\}, \{1, 5\}, \{2, 3\}, \{3, 4\}, \{4, 5\}\} \quad \textbf{(10)}$$

```
> seq( x[e[1]]^k-x[e[2]]^k, e=Edges(W5) );
```

$$x_0^k - x_1^k,\ x_0^k - x_2^k,\ x_0^k - x_3^k,\ x_0^k - x_4^k,\ x_0^k - x_5^k,\ x_1^k - x_2^k,\ x_1^k - x_5^k,\ x_2^k - x_3^k,\ x_3^k - x_4^k,\ x_4^k - x_5^k \quad \textbf{(11)}$$

```
> Sys := proc(G::Graph,x::name,k::nonnegint)
  local V,E,v,e,S;
     V,E := Vertices(G), Edges(G);
     S := {seq( x[v]^k-1, v=V )} union
          {seq( normal( (x[e[1]]^k-x[e[2]]^k) / (x[e[1]]-x[e[2]])
  ), e=E )}
  end:
> S := Sys( WheelGraph(3), x, 3 );
```

$$S := \{x_0^3 - 1,\ x_1^3 - 1,\ x_2^3 - 1,\ x_3^3 - 1,\ x_0^2 + x_1 x_0 + x_1^2,\ x_0^2 + x_2 x_0 + x_2^2,\ x_0^2 + x_3 x_0 + x_3^2,\ x_1^2 + x_2 x_1 + x_2^2,\ x_1^2 + x_3 x_1 + x_3^2,\ x_2^2 + x_3 x_2 + x_3^2\} \quad \textbf{(12)}$$

```
> S := subs( x[0]=1, S );
```

$$S := \{0,\ x_1^3 - 1,\ x_2^3 - 1,\ x_3^3 - 1,\ 1 + x_1 + x_1^2,\ 1 + x_2 + x_2^2,\ 1 + x_3 + x_3^2,\ x_1^2 + x_2 x_1 + x_2^2,\ x_1^2 \quad \textbf{(13)}$$

$+x_3\,x_1+x_3^2,\; x_2^2+x_3\,x_2+x_3^2\}$

```
> Groebner[Basis]( S, tdeg(x[0],x[1],x[2],x[3]) );
```

$$[1] \tag{14}$$

```
> S := Sys( WheelGraph(4), x, 3 );
```

$$S := \{x_0^3-1,\; x_1^3-1,\; x_2^3-1,\; x_3^3-1,\; x_4^3-1,\; x_0^2+x_1\,x_0+x_1^2,\; x_0^2+x_2\,x_0+x_2^2,\; x_0^2+x_3\,x_0+x_3^2,\; x_0^2+x_4\,x_0+x_4^2,\; x_1^2+x_2\,x_1+x_2^2,\; x_1^2+x_4\,x_1+x_4^2,\; x_2^2+x_3\,x_2+x_3^2,\; x_3^2+x_4\,x_3+x_4^2\} \tag{15}$$

```
> S := subs( x[0]=1, S );
```

$$S := \{0,\; x_1^3-1,\; x_2^3-1,\; x_3^3-1,\; x_4^3-1,\; 1+x_1+x_1^2,\; 1+x_2+x_2^2,\; 1+x_3+x_3^2,\; 1+x_4+x_4^2,\; x_1^2+x_2\,x_1+x_2^2,\; x_1^2+x_4\,x_1+x_4^2,\; x_2^2+x_3\,x_2+x_3^2,\; x_3^2+x_4\,x_3+x_4^2\} \tag{16}$$

```
> Groebner[Basis](S, tdeg(x[1],x[2],x[3],x[4]) );
```

$$[x_3+1+x_4,\; x_2-x_4,\; x_1+1+x_4,\; 1+x_4+x_4^2] \tag{17}$$

```
> Monomials := proc(X::set(name),d::nonnegint) option remember;
  local x,i,m;
     # return all monomials in X of degree 0,1,...,d .
     if X={} then return 1 fi;
     x := X[1];
     seq( seq( x^i*m, m=Monomials( X[2..-1], d-i ) ), i=0..d );
  end:
> Monomials( {x,y,z}, 3 );
```

$$1,\; z,\; z^2,\; z^3,\; y,\; y\,z,\; y\,z^2,\; y^2,\; y^2\,z,\; y^3,\; x,\; x\,z,\; x\,z^2,\; x\,y,\; x\,y\,z,\; x\,y^2,\; x^2,\; x^2\,z,\; x^2\,y,\; x^3 \tag{18}$$

```
> HNSS := proc( S::set(polynom), d::nonnegint, c::name,
  ansatz::name )
  local M,i,one,X;
      X := indets(S,name); #print(X);
      M := [Monomials(X,d)]; #print(M);
      one := add( add( c[i,j]*M[j], j=1..nops(M) )*S[i], i=1..nops
  (S) );
      if nargs=4 then ansatz := one fi;
      { coeffs( expand(one)-1, X ) };
  end:
```

Consider W(5)

```
> S := HNSS( Sys(WheelGraph(5),x,3), 1, c, 'ANS' );
```

$$S := \{c_{1,2},\; c_{1,3},\; c_{1,4},\; c_{1,5},\; c_{1,6},\; c_{1,7},\; c_{2,2},\; c_{2,3},\; c_{2,4},\; c_{2,5},\; c_{2,6},\; c_{2,7},\; c_{3,2},\; c_{3,3},\; c_{3,4},\; c_{3,5},\; c_{3,6},\; c_{3,7},\; c_{4,2},\; c_{4,3},\; c_{4,4},\; c_{4,5},\; c_{4,6},\; c_{4,7},\; c_{5,2},\; c_{5,3},\; c_{5,4},\; c_{5,5},\; c_{5,6},\; c_{5,7},\; c_{6,2},\; c_{6,3},\; c_{6,4},\; c_{6,5},\; c_{6,6},\; c_{6,7},\; c_{7,1},\; c_{8,1},\; c_{9,1},\; c_{10,1},\; c_{11,1},\; c_{12,1},\; c_{12,3},\; c_{13,1},\; c_{13,4},\; c_{14,1},\; c_{14,2},\; c_{15,1},\; c_{15,6},\; c_{16,1},\; c_{16,5},\; c_{7,3}+c_{10,6},\; c_{8,2}+c_{11,5},\; c_{9,2}+c_{11,4},\; c_{9,6}+c_{7,4}, \tag{19}$$

$$c_{10,5}+c_{8,3},\ c_{12,2}+c_{13,5},\ c_{12,4}+c_{14,6},\ c_{13,3}+c_{16,6},\ c_{15,5}+c_{14,3},\ c_{16,4}+c_{15,2},$$
$$c_{7,1}+c_{12,1}+c_{13,1},\ c_{8,4}+c_{9,5}+c_{14,7},\ c_{8,6}+c_{12,7}+c_{7,5},\ c_{9,1}+c_{15,1}+c_{14,1},\ c_{9,3}$$
$$+c_{15,7}+c_{10,4},\ c_{9,6}+c_{15,6}+c_{14,6},\ c_{10,1}+c_{16,1}+c_{15,1},\ c_{10,2}+c_{11,3}+c_{16,7},$$
$$c_{11,1}+c_{13,1}+c_{16,1},\ c_{11,4}+c_{16,4}+c_{13,4},\ c_{12,3}+c_{8,3}+c_{14,3},\ c_{12,3}+c_{13,3}+c_{7,3},$$
$$c_{12,4}+c_{13,4}+c_{7,4},\ c_{13,7}+c_{7,2}+c_{11,6},\ c_{14,1}+c_{12,1}+c_{8,1},\ c_{14,2}+c_{12,2}+c_{8,2},$$
$$c_{15,2}+c_{14,2}+c_{9,2},\ c_{16,5}+c_{10,5}+c_{15,5},\ c_{16,5}+c_{13,5}+c_{11,5},\ c_{16,6}+c_{10,6}$$
$$+c_{15,6},\ c_{2,1}+c_{12,6}+c_{13,6}+c_{7,6},\ c_{7,5}+c_{13,5}+c_{12,5}+c_{12,6},\ c_{7,7}+c_{13,7}+c_{7,6}$$
$$+c_{12,7},\ c_{8,4}+c_{14,4}+c_{12,4}+c_{14,5},\ c_{8,5}+c_{14,7}+c_{12,7}+c_{8,7},\ c_{9,3}+c_{15,4}+c_{15,3}$$
$$+c_{14,3},\ c_{9,4}+c_{14,7}+c_{9,7}+c_{15,7},\ c_{10,2}+c_{16,2}+c_{15,2}+c_{16,3},\ c_{10,3}+c_{10,7}$$
$$+c_{15,7}+c_{16,7},\ c_{10,3}+c_{16,3}+c_{5,1}+c_{15,3},\ c_{11,7}+c_{11,2}+c_{16,7}+c_{13,7},\ c_{13,2}$$
$$+c_{6,1}+c_{11,2}+c_{16,2},\ c_{13,2}+c_{13,6}+c_{11,6}+c_{16,6},\ c_{13,6}+c_{12,2}+c_{7,2}+c_{13,2},$$
$$c_{14,5}+c_{12,5}+c_{3,1}+c_{8,5},\ c_{14,6}+c_{8,6}+c_{12,6}+c_{12,5},\ c_{15,4}+c_{4,1}+c_{14,4}+c_{9,4},$$
$$c_{15,5}+c_{14,4}+c_{14,5}+c_{9,5},\ c_{16,3}+c_{16,2}+c_{13,3}+c_{11,3},\ c_{16,4}+c_{10,4}+c_{15,3}$$
$$+c_{15,4},\ c_{11,1}+c_{7,1}+c_{9,1}+c_{8,1}+c_{10,1},\ -c_{1,2}-c_{5,2}-c_{3,2}-c_{6,2}-c_{4,2}-c_{2,2},$$
$$-c_{2,5}-c_{5,5}-c_{4,5}-c_{6,5}-c_{1,5}-c_{3,5},\ -c_{3,3}-c_{6,3}-c_{4,3}-c_{5,3}-c_{2,3}-c_{1,3},$$
$$-c_{4,6}-c_{3,6}-c_{1,6}-c_{5,6}-c_{6,6}-c_{2,6},\ -c_{6,4}-c_{5,4}-c_{3,4}-c_{4,4}-c_{2,4}-c_{1,4},$$
$$-c_{6,7}-c_{4,7}-c_{5,7}-c_{3,7}-c_{1,7}-c_{2,7},\ c_{7,2}+c_{10,2}+c_{9,2}+c_{11,2}+c_{8,2}+c_{11,7},$$
$$c_{7,6}+c_{9,6}+c_{8,6}+c_{10,6}+c_{7,7}+c_{11,6},\ c_{9,4}+c_{7,4}+c_{11,4}+c_{9,7}+c_{10,4}+c_{8,4},$$
$$c_{10,5}+c_{8,5}+c_{8,7}+c_{11,5}+c_{9,5}+c_{7,5},\ c_{10,7}+c_{10,3}+c_{11,3}+c_{8,3}+c_{9,3}+c_{7,3},$$
$$c_{10,7}+c_{11,7}+c_{7,7}+c_{1,1}+c_{9,7}+c_{8,7},\ -1-c_{1,1}-c_{2,1}-c_{3,1}-c_{4,1}-c_{5,1}$$
$$-c_{6,1}\}$$

```
> ANS;
```

$$\left(c_{1,1}+c_{1,2}x_5+c_{1,3}x_4+c_{1,4}x_3+c_{1,5}x_2+c_{1,6}x_1+c_{1,7}x_0\right)\left(x_0^3-1\right)+\left(c_{2,1}\right.$$
$$+c_{2,2}x_5+c_{2,3}x_4+c_{2,4}x_3+c_{2,5}x_2+c_{2,6}x_1+c_{2,7}x_0\left.\right)\left(x_1^3-1\right)+\left(c_{3,1}\right.$$
$$+c_{3,2}x_5+c_{3,3}x_4+c_{3,4}x_3+c_{3,5}x_2+c_{3,6}x_1+c_{3,7}x_0\left.\right)\left(x_2^3-1\right)+\left(c_{4,1}\right.$$
$$+c_{4,2}x_5+c_{4,3}x_4+c_{4,4}x_3+c_{4,5}x_2+c_{4,6}x_1+c_{4,7}x_0\left.\right)\left(x_3^3-1\right)+\left(c_{5,1}\right.$$
$$+c_{5,2}x_5+c_{5,3}x_4+c_{5,4}x_3+c_{5,5}x_2+c_{5,6}x_1+c_{5,7}x_0\left.\right)\left(x_4^3-1\right)+\left(c_{6,1}\right.$$
$$+c_{6,2}x_5+c_{6,3}x_4+c_{6,4}x_3+c_{6,5}x_2+c_{6,6}x_1+c_{6,7}x_0\left.\right)\left(x_5^3-1\right)+\left(c_{7,1}\right. \tag{20}$$

$$+ c_{7,2} x_5 + c_{7,3} x_4 + c_{7,4} x_3 + c_{7,5} x_2 + c_{7,6} x_1 + c_{7,7} x_0) (x_0^2 + x_1 x_0 + x_1^2) + (c_{8,1}$$
$$+ c_{8,2} x_5 + c_{8,3} x_4 + c_{8,4} x_3 + c_{8,5} x_2 + c_{8,6} x_1 + c_{8,7} x_0) (x_0^2 + x_2 x_0 + x_2^2) + (c_{9,1}$$
$$+ c_{9,2} x_5 + c_{9,3} x_4 + c_{9,4} x_3 + c_{9,5} x_2 + c_{9,6} x_1 + c_{9,7} x_0) (x_0^2 + x_3 x_0 + x_3^2)$$
$$+ (c_{10,1} + c_{10,2} x_5 + c_{10,3} x_4 + c_{10,4} x_3 + c_{10,5} x_2 + c_{10,6} x_1 + c_{10,7} x_0) (x_0^2 + x_4 x_0$$
$$+ x_4^2) + (c_{11,1} + c_{11,2} x_5 + c_{11,3} x_4 + c_{11,4} x_3 + c_{11,5} x_2 + c_{11,6} x_1 + c_{11,7} x_0) (x_0^2$$
$$+ x_5 x_0 + x_5^2) + (c_{12,1} + c_{12,2} x_5 + c_{12,3} x_4 + c_{12,4} x_3 + c_{12,5} x_2 + c_{12,6} x_1$$
$$+ c_{12,7} x_0) (x_1^2 + x_2 x_1 + x_2^2) + (c_{13,1} + c_{13,2} x_5 + c_{13,3} x_4 + c_{13,4} x_3 + c_{13,5} x_2$$
$$+ c_{13,6} x_1 + c_{13,7} x_0) (x_1^2 + x_5 x_1 + x_5^2) + (c_{14,1} + c_{14,2} x_5 + c_{14,3} x_4 + c_{14,4} x_3$$
$$+ c_{14,5} x_2 + c_{14,6} x_1 + c_{14,7} x_0) (x_2^2 + x_3 x_2 + x_3^2) + (c_{15,1} + c_{15,2} x_5 + c_{15,3} x_4$$
$$+ c_{15,4} x_3 + c_{15,5} x_2 + c_{15,6} x_1 + c_{15,7} x_0) (x_3^2 + x_4 x_3 + x_4^2) + (c_{16,1} + c_{16,2} x_5$$
$$+ c_{16,3} x_4 + c_{16,4} x_3 + c_{16,5} x_2 + c_{16,6} x_1 + c_{16,7} x_0) (x_4^2 + x_5 x_4 + x_5^2)$$

Try to find a degree 1 certificate for W(5)

```
> nops(S);
```

115 **(21)**

```
> nops( indets(S) );
```

112 **(22)**

```
> {solve(S)};
```

{ } **(23)**

```
> S := HNSS( Sys(WheelGraph(5),x,3), 2, c ):
```

A degree 2 certificate

```
> nops(S);
```

336 **(24)**

```
> nops( indets(S) );
```

448 **(25)**

```
> {solve(S)};
```

{ } **(26)**

```
> S := HNSS( Sys(WheelGraph(5),x,3), 3, c ):
```

A degree 3 certificate

```
> nops(S);
```

783 **(27)**

```
> nops(indets(S));
```

1344 **(28)**

```
> sol := {solve(S)}: # should output {} but getting bad answer in
```

Maple 17

```
> S := HNSS( Sys(WheelGraph(5),x,3), 4, c, 'ZZ' ):
```

A degree 4 certificate

```
> nops(S);
```

1590 **(29)**

```
> nops(indets(S));
```

3360 **(30)**

```
> sol := solve(S);
```

[Length of output exceeds limit of 1000000] **(31)**

```
> Certificate := proc(ansatz, sol, c) local p;
    p := map( proc(e) if lhs(e)=rhs(e) then lhs(e)=1 fi end, sol
  );
    subs( p, subs( sol, ansatz ) );
  end:
> Certificate( ZZ, sol, c ):
> expand(%);
```

1 **(32)**

```
> S := HNSS( Sys(WheelGraph(5),x,3), 1, c, 'ZZ' ):
```

But a degree 1 certificate exists modulo 2 !!

```
> sol := msolve( S, 2 );
```

(33)

$sol := \{c_{1,1} = \_Z5 + \_Z3 + \_Z1 + \_Z4 + \_Z2, c_{1,2} = 0, c_{1,3} = 0, c_{1,4} = 0, c_{1,5} = 0, c_{1,6} = 0, c_{1,7} = 0, c_{2,1} = \_Z7 + \_Z6 + \_Z10 + \_Z12 + \_Z15 + \_Z1, c_{2,2} = 0, c_{2,3} = 0, c_{2,4} = 0, c_{2,5} = 0, c_{2,6} = 0, c_{2,7} = 0, c_{3,1} = \_Z9 + \_Z2 + 1 + \_Z6 + \_Z10, c_{3,2} = 0, c_{3,3} = 0, c_{3,4} = 0, c_{3,5} = 0, c_{3,6} = 0, c_{3,7} = 0, c_{4,1} = \_Z11 + 1 + \_Z9 + \_Z12 + \_Z3, c_{4,2} = 0, c_{4,3} = 0, c_{4,4} = 0, c_{4,5} = 0, c_{4,6} = 0, c_{4,7} = 0, c_{5,1} = \_Z15 + \_Z4 + 1 + \_Z11 + \_Z13, c_{5,2} = 0, c_{5,3} = 0, c_{5,4} = 0, c_{5,5} = 0, c_{5,6} = 0, c_{5,7} = 0, c_{6,1} = \_Z5 + \_Z13 + \_Z7, c_{6,2} = 0, c_{6,3} = 0, c_{6,4} = 0, c_{6,5} = 0, c_{6,6} = 0, c_{6,7} = 0, c_{7,1} = 0, c_{7,2} = \_Z14 + 1, c_{7,3} = \_Z14, c_{7,4} = \_Z10 + \_Z12 + \_Z15 + \_Z14, c_{7,5} = \_Z12 + \_Z15 + \_Z14 + \_Z10 + 1, c_{7,6} = \_Z10 + \_Z12 + \_Z15 + \_Z1, c_{7,7} = \_Z1, c_{8,1} = 0, c_{8,2} = \_Z8 + \_Z14, c_{8,3} = \_Z12 + \_Z15 + \_Z14, c_{8,4} = 1 + \_Z12 + \_Z15 + \_Z14, c_{8,5} = \_Z2 + \_Z8 + \_Z12 + \_Z15, c_{8,6} = 1 + \_Z8 + \_Z14, c_{8,7} = \_Z2, c_{9,1} = 0, c_{9,2} = \_Z15 + \_Z14, c_{9,3} = \_Z14 + 1 + \_Z15, c_{9,4} = \_Z12 + \_Z3 + \_Z10, c_{9,5} = \_Z12 + \_Z15 + \_Z14 + \_Z10 + 1, c_{9,6} = \_Z10 + \_Z12 + \_Z15 + \_Z14, c_{9,7} = \_Z3, c_{10,1} = 0, c_{10,2} = \_Z14 + 1, c_{10,3} = \_Z15 + \_Z12 + \_Z4, c_{10,4} = 1 + \_Z12 + \_Z15 + \_Z14, c_{10,5} = \_Z12 + \_Z15 + \_Z14, c_{10,6} = \_Z14, c_{10,7}$

$= \_Z4,\ c_{11,1} = 0,\ c_{11,2} = \_Z15 + \_Z8 + \_Z5,\ c_{11,3} = \_Z14 + 1 + \_Z15,\ c_{11,4}$
$= \_Z15 + \_Z14,\ c_{11,5} = \_Z8 + \_Z14,\ c_{11,6} = 1 + \_Z8 + \_Z14,\ c_{11,7} = \_Z5,\ c_{12,1}$
$= 0,\ c_{12,2} = \_Z8 + \_Z14,\ c_{12,3} = 0,\ c_{12,4} = \_Z10 + \_Z12 + \_Z15 + \_Z14,\ c_{12,5}$
$= 1 + \_Z6 + \_Z12 + \_Z10 + \_Z15 + \_Z8,\ c_{12,6} = \_Z6,\ c_{12,7} = \_Z8 + \_Z10 + \_Z12$
$+ \_Z15,\ c_{13,1} = 0,\ c_{13,2} = 1 + \_Z7 + \_Z8,\ c_{13,3} = \_Z14,\ c_{13,4} = 0,\ c_{13,5} = \_Z8$
$+ \_Z14,\ c_{13,6} = \_Z7,\ c_{13,7} = \_Z8,\ c_{14,1} = 0,\ c_{14,2} = 0,\ c_{14,3} = \_Z12 + \_Z15$
$+ \_Z14,\ c_{14,4} = \_Z10 + 1 + \_Z9,\ c_{14,5} = \_Z9,\ c_{14,6} = \_Z10 + \_Z12 + \_Z15$
$+ \_Z14,\ c_{14,7} = \_Z10,\ c_{15,1} = 0,\ c_{15,2} = \_Z15 + \_Z14,\ c_{15,3} = 1 + \_Z11 + \_Z12,$
$c_{15,4} = \_Z11,\ c_{15,5} = \_Z12 + \_Z15 + \_Z14,\ c_{15,6} = 0,\ c_{15,7} = \_Z12,\ c_{16,1} = 0,$
$c_{16,2} = \_Z13 + 1 + \_Z15,\ c_{16,3} = \_Z13,\ c_{16,4} = \_Z15 + \_Z14,\ c_{16,5} = 0,\ c_{16,6}$
$= \_Z14,\ c_{16,7} = \_Z15\}$

```
> Zvals := map( proc(e) if type(rhs(e),name) then rhs(e) fi end,
  sol );
```

$Zvals := \{\_Z1, \_Z10, \_Z11, \_Z12, \_Z13, \_Z14, \_Z15, \_Z2, \_Z3, \_Z4, \_Z5, \_Z6, \_Z7, \_Z8, \_Z9\}$ **(34)**

```
> one := subs( {seq( z=1, z=Zvals)}, subs( sol, ZZ ) ) mod 2;
```

$one := x_0^3 + 1 + x_2^3 + x_3^3 + x_4^3 + x_5^3 + (x_4 + x_2 + x_0)(x_0^2 + x_1 x_0 + x_1^2) + (x_4 + x_1 + x_0)(x_0^2 + x_2 x_0 + x_2^2) + (x_4 + x_3 + x_2 + x_0)(x_0^2 + x_3 x_0 + x_3^2) + (x_4 + x_2 + x_1 + x_0)(x_0^2 + x_4 x_0 + x_4^2) + (x_5 + x_4 + x_1 + x_0)(x_0^2 + x_5 x_0 + x_5^2) + x_1 (x_1^2 + x_2 x_1 + x_2^2) + (x_5 + x_4 + x_1 + x_0)(x_1^2 + x_5 x_1 + x_5^2) + (x_4 + x_3 + x_2 + x_0)(x_2^2 + x_3 x_2 + x_3^2) + (x_4 + x_3 + x_2 + x_0)(x_3^2 + x_4 x_3 + x_4^2) + (x_5 + x_4 + x_1 + x_0)(x_4^2 + x_5 x_4 + x_5^2)$ **(35)**

```
> Expand(%) mod 2;
```

1 **(36)**